
pkgname=libgudev
pkgver=232
pkgrel=3
pkgdesc="UDev GObject bindings"
arch=('x86_64')
license=('LGPL')
url="https://git.gnome.org/browse/libgudev"
depends=('systemd' 'glib2')
makedepends=('gobject-introspection')
source=("https://ftp.gnome.org/pub/gnome/sources/libgudev/${pkgver}/${pkgname}-${pkgver}.tar.xz")
#source=("https://github.com/systemd-devs/libgudev/archive/$pkgname-$pkgver.tar.gz")
sha256sums=('ee4cb2b9c573cdf354f6ed744f01b111d4b5bed3503ffa956cefff50489c7860')

build() {
  cd ${pkgname}-${pkgver}

  ./configure --prefix=/usr --disable-umockdev
  make
}

check() {
  cd ${pkgname}-${pkgver}

  make check
}

package() {
  cd ${pkgname}-${pkgver}

  make DESTDIR=${pkgdir} install
}
